


# FUZE BASIC

## Worksheet: 11a

### "Seven Segment LED"

Seven segment LEDs or multiple digit versions are used all over the place. Take a look at your washing machine, microwave, digital clock and so on. Ever wondered how they work. Well as ever with these things it's rarely as complicated as you might expect. Please go to immediate mode.




Copy the wiring image to the right on your breadboard. Notice the seven segment LED has a dot on the bottom left hand corner. This is actually a decimal point but it also shows us which way round to put it.

The **RED** wire goes from **3.3V** to the middle pin along the bottom. The **BLUE** wire goes from **GPIO** pin 0 to the first pin on the left along the bottom.

The **3.3V** is, inside the LED, connected to one end of each of the segments. To make one light up we therefore need to send a Ground signal to the segment.

**Please note** the middle top pin should not be connected. If you do you might reset the **FUZE** which, is not actually that bad, but you will have to wait for it to reboot.


In immediate mode type in;



```

Pinmode ( 0, 1 )
Digitalwrite ( 0, 0 )
    
```

Notice when you enter the **DIGITALWRITE (0, 0)** command the bottom left segment lights up. The 0 is the same as a ground signal so the segment circuit is completed and it lights up.



Unplug the **BLUE** wire from the **LED** and try connecting it to the other points (just not the one with a **RED X!**). You will see which point connects to which segment.


Now connect up the breadboard as displayed on the right. We don't need to connect **GPIO** pin 7 at this point.

```

For pin = 0 To 6 Loop
pinMode (pin , 1)
Repeat
Loop
For pin = 0 To 6 Loop
digitalWrite (pin, 0)
Repeat
Wait (.3)
For pin = 0 To 6 Loop
digitalWrite (pin, 1)
Repeat
Wait (.3)
Repeat
    
```

Go to the **Editor (F2)** and type in the program on the left.

The first **LOOP** sets 7 **GPIO** pins to output mode. Then we run two **LOOPS**, one sending a low signal, turning the segments ON. The second sends a high signal switching them off.



**Challenge:** See if you can make the LEDs snake around the display rather than all going on and off at the same time. Hint: Move the **WAIT** command - perhaps make it faster too!

In the **Editor** enter the program as displayed on the right;

As you can see we are directly controlling all of the segments individually. In this case a 0 is turning every segment on. It also looks the same as an 8.

Try changing some of the 0's to 1's and see if you can make other different numbers to be displayed. For example try 1, 0, 0, 0, 1, 0, 0. This will display the number 3 - clever eh!

It would be very messy to list all the **DIGITALWRITE** commands for the digits 0 to 9. What we need is some kind of matrix to store everything in so we can access it quickly and easily. To do this we use a special variable called an **ARRAY**.

In the **Editor** enter this program;

```

Proc setup
Loop
Ink = Raspberry
Printat (0, 0); "Press 0-9"
key = Inkey - 48
If key >= 0 And key <= 9 Then
For segment = 0 To 6 Loop
Digitalwrite (segment, led(key, segment))
Repeat
Endif
Repeat
Def proc setup
For pin = 0 To 6 Loop
Pinmode (pin, 1)
Repeat
Fontsize (14)
Dim led(9, 6)
For number = 0 To 9 Loop
For segment = 0 To 6 Loop
Read onoff
led(number, segment) = onoff
Repeat
Repeat
Data 0, 0, 0, 1, 0, 0, 0
Data 1, 1, 0, 1, 1, 1, 0
Data 0, 0, 1, 0, 1, 0, 0
Data 1, 0, 0, 0, 1, 0, 0
Data 1, 1, 0, 0, 0, 1, 0
Data 1, 0, 0, 0, 0, 0, 1
Data 0, 0, 0, 0, 0, 0, 1
Data 1, 1, 0, 1, 1, 0, 0
Data 0, 0, 0, 0, 0, 0, 0
Data 1, 0, 0, 0, 0, 0, 0
Cls
Endproc
    
```

This needs some explanation. The first line **PROC SETUP** jumps to a routine called **SETUP (DEF PROC SETUP)**. The sets the **GPIO** pins for output and the **FONTSIZE**.

The command **DIM LED (9, 6)** initialises a new **ARRAY** variable called **LED** and gives it an index of 0 to 9 down and 0 to 6 across. We use two **FOR LOOPS** to read a sequence of **DATA** values and store these in the **ARRAY** variable **LED**. In other words the variable **LED** now looks like this chart;

Now when we want to display a digit we just need to look up the arrangement of on's and off's in the **ARRAY** so **number 4** looks like "**1100010**" or "**off off on on on on off on**" - remember in this case the off is sending a ground signal which switches the **segment** on.

In our program we're using a system variable that always has the code value of the current key being pressed stored in it. The variable, **INKEY** can be read to see if a key is pressed. The code value for the key **0** on your keyboard is **48**, **1 is 49**, **2 is 50** and so on. If we subtract **48** from the reading we're left with the values **0 to 9**. Next we check to make sure it is only **0 to 9** and **IF** so, display the key on screen and run the **LOOP** to set the segments on or off as defined in the **LED ARRAY** with the statement; **DIGITALWRITE (SEGMENT, LED (KEY, SEGMENT))**.

The first use of **SEGMENT** here is being used for the **PIN** number also. Notice the use of the **ARRAY** variable **LED (KEY, SEGMENT)**.

**Challenge:** This is your toughest challenge yet, but if you can do it then you have nailed **ARRAYS**. Given that the code for the letter 'a' is **97** could you modify the program to display letters instead of numbers. **Hints:** use **0 to 25** for the alphabet - define capital letters on the **LED** not lower case - don't forget to change the **LED ARRAY** to **(25, 6)** to support the full alphabet and increase the **NUMBER LOOP** to **25**. **Good luck** and don't forget to change, and add, the **DATA** values from 10 digits to 26 alphabet characters.

```

For pin = 0 to 7 Loop
Pinmode ( pin, 1 )
Repeat
Loop
Digitalwrite ( 0, 0 )
Digitalwrite ( 1, 0 )
Digitalwrite ( 2, 0 )
Digitalwrite ( 3, 0 )
Digitalwrite ( 4, 0 )
Digitalwrite ( 5, 0 )
Digitalwrite ( 6, 0 )
Repeat
    
```

**DIM LED (9, 6)** creates a new **ARRAY** variable indexed as shown here;

		SEGMENT						
		0	1	2	3	4	5	6
NUMBER	0	0	0	0	1	0	0	0
	1	1	1	0	1	1	1	0
	2	0	0	1	0	1	0	0
	3	1	0	0	0	1	0	0
	4	1	1	0	0	0	1	0
	5	1	0	0	0	0	0	1
	6	0	0	0	0	0	0	1
	7	1	1	0	1	1	0	0
	8	0	0	0	0	0	0	0
	9	1	0	0	0	0	0	0